

Maple: Solving Ordinary Differential Equations

A differential equation is an equation that involves derivatives of one or more unknown functions. Solving the differential equation means finding a function (or every such function) that satisfies the differential equation. Many of the fundamental laws of physics, chemistry, biology and economics can be formulated as differential equations. Differential equations are often classified with respect to order. The order of a differential equation is the order of the highest derivative present in the equation. An ordinary differential equation (ODE) is a differential equation in which the unknown function in question is a function of a single independent variable. We shall only look at first and second order ODEs in this chapter.

Obtaining the General Solution of a First Order ODE

We will solve the following first order ODE,

$$\frac{dy}{dx} = 2xy, \quad y(0) = 2. \quad (1)$$

This is a rather straight forward ODE but will allow us to demonstrate the approach to solving ODEs using Maple. Note that equation (1) is an initial value problem. This means that we are attempting to find an exact solution depending on a particular initial condition, in this case $y(0) = 2$: the value of y , when x is 0, is 2. We will first solve the problem ignoring the initial condition and obtain a general solution to the problem. Before attempting to solve an ODE in Maple it is necessary to “load” particular commands and functions that will be needed. The commands for dealing with ODEs in Maple are stored in the package “DEtools”. Additional commands for plotting ODE solution curves can be found in the package “plots”. Therefore our first step is to load these two packages using the `with` command:

```
> with(plots):  
> with(DEtools):
```

Maple: Solving Ordinary Differential Equations

The next step is to input the ODE that we are attempting to solve. Remember that the function y depends on x and so it is necessary to input it as $y(x)$ so that Maple is able to recognise the dependency. We shall label equation (1) as ODE1 using the assignment operator:

```
> ODE1:=diff(y(x),x)=2*x*y(x);
```

$$ODE1 := \frac{d}{dx} y(x) = 2xy(x)$$

The command for solving an ODE is `dsolve`. Remember that if you are unfamiliar with any command you can look up the help file associated with it by simply placing a question mark (?) in front of the command and pressing enter. We shall now solve equation (1) and attempt to obtain the general solution.


```
> dsolve(ODE1,y(x));
```

$$y(x) = _C1 e^{x^2}$$

This is the general solution of equation (1). Note the “`_C1`” is Maple’s way of representing an arbitrary constant. In more complicated solutions this arbitrary constant might appear after the term that it is associated with.

It is also possible to plot the solution curves of the general solution. To see a direction field plot of the solution curves of equation (1) use the following command.

```
> dfieldplot(ODE1,y(x),x=-2..2,y=-2..2,color=blue,scaling=constrained,  
> dirgrid=[40,40]);
```



The first entry in parentheses is the differential equation ODE1. The second entry names the dependent variable $y(x)$. The third and fourth entries give ranges for the independent and

dependent variables: $x = -2 \dots 2, y = -2 \dots 2$. The remaining entries are options and can be omitted if desired. It is common practice that whenever you are plotting direction fields you use the “scaling = constrained” option as otherwise the graph will appear misleading because the direction lines will be distorted.


Obtaining the Exact Solution of a First Order ODE

We can take an initial condition into account using Maple and determine an exact solution that depends on the initial condition. Solving an ODE that has an initial condition is again done using the `dsolve` command. Using the initial condition presented in equation (1) we shall now determine an exact solution to the ODE.

```
> dsolve({ODE1, y(0)=2}, y(x));  
y(x) = 2 e(x2)
```

If we wish to plot a particular solution curve then we use the command `DEplot`. `DEplot` will plot the direction field lines as well as the particular solution line depending on the initial condition.

```
> DEplot(ODE1, y(x), -2..2, [y(0)=2], linecolor=magenta, color=blue,  
> arrows=LINE);
```



If you do not want the direction field plotted with the particular solution, you use the option “`arrows=NONE`” in the `DEplot` command.

Solving Second Order ODEs using Maple

We will start by solving the following second order constant coefficient homogeneous equation.

$$y'' + 2y' + 10y = 0.$$

This equation is homogeneous because all the terms that involve the unknown function y and its derivatives appear on the left hand side of the equation and the right hand side is zero. We will start by entering the above equation into Maple. Recall that the packages “DEtools” and “plots” will also be required to solve second order ODEs and so will have to be added at the start as we did in the First Order example.

```
> eq1 := diff(y(t),t,t) + 2*diff(y(t),t) + 10*y(t) = 0;
      eq1 := (d^2/dt^2 y(t)) + 2(d/dt y(t)) + 10y(t) = 0
```

Recall that `diff(y(t),t,t)` will give you the second derivative of $y(t)$. We will again use the `dsolve` command to solve the differential equation. The first argument is the differential equation that we are solving (`eq1`) and the second is the function to be found ($y(t)$). Recall from the previous example that when we use the `dsolve` command it gives us a solution of the form “unknown variable = solution”. If we only wish to see the solution we can use the `rhs` command to display the right hand side only. We shall save the solution in the variable “sol1”.

```
> sol1 := rhs(dsolve(eq1,y(t)));
      sol1 := _C1 e^(-t) sin(3 t) + _C2 e^(-t) cos(3 t)
```

Recall that “_C1” and “_C2” are Maple’s way of representing arbitrary constants.

Solving an Initial Value Second Order ODE

The next step will be to solve a second order ODE that includes initial conditions. Every second order ODE will have two initial conditions. We shall now solve the original second order ODE based on the following initial conditions:


$$y'' + 2y' + 10y = 0, \quad y(0) = 3, y'(0) = -5.$$

Use the `dsolve` command to solve the ODE based on the initial conditions. Save the solution in the variable `sol2`.

```
> sol2 := rhs(dsolve({eq1,y(0)=3,D(y)(0)=-5},y(t)));
      sol2 := -2/3 e^(-t) sin(3 t) + 3 e^(-t) cos(3 t)
```

We now have the exact solution of the equation. Plotting the solution might make it easier to understand what is going on. We simply use the `plot` command to plot `sol2`.

```
> plot(sol2,t=-1..6,labels=["t","y"]);
```



Inhomogeneous Second Order ODE with ICs

We shall now look at how to solve an inhomogeneous second order ODE with initial conditions. Consider the following problem:

$$y'' + y' + y = t^2 \cos(2t), \quad y(0) = 0, y'(0) = 2.$$

As before our first step is to enter the equation into Maple ignoring the initial conditions. We shall label our new ODE as **eq2**.

```
> eq2 := diff(y(t),t,t) + diff(y(t),t) + y(t) = t^2*cos(2*t);
```

$$eq2 := \left(\frac{d^2}{dt^2} y(t)\right) + \left(\frac{d}{dt} y(t)\right) + y(t) = t^2 \cos(2t)$$

We can again solve this second order ODE using the **dsolve** command. We shall call our solution **sol3**.

```
> sol3 := rhs(dsolve(eq2,y(t)));
```

$$sol3 := e^{(-\frac{t}{2})} \sin\left(\frac{\sqrt{3}t}{2}\right) - C2 + e^{(-\frac{t}{2})} \cos\left(\frac{\sqrt{3}t}{2}\right) - C1 + \frac{1}{2197} (338 t^2 + 832 t - 1212) \sin(2 t) + \frac{1}{2197} \cos(2 t) (336 - 507 t^2 + 1118 t)$$

This solution is more complex than in the previous example due to the inhomogeneous terms on the right hand side of the problem. We now solve the initial value problem taking into account our initial conditions. Again this is done quite easily using the `dsolve` command. We label the solution of the initial value problem `sol4`.

```
> sol4 := rhs(dsolve({eq2,y(0)=0,D(y)(0)=2},y(t)));
```

$$\begin{aligned} \text{sol4} := & \frac{3688}{2197} e^{(-\frac{t}{2})} \sin\left(\frac{\sqrt{3}t}{2}\right) \sqrt{3} - \frac{336}{2197} e^{(-\frac{t}{2})} \cos\left(\frac{\sqrt{3}t}{2}\right) \\ & + \frac{1}{2197} (338t^2 + 832t - 1212) \sin(2t) + \frac{1}{2197} \cos(2t) (336 - 507t^2 + 1118t) \end{aligned}$$

Even taking the initial conditions into account this answer is quite complicated and so a plot of the above function might make it easier to see the overall nature of the solution curve.

```
> plot(sol4,t=0..18,labels=["t","y"]);
```

